



Application Note		資料作成 090309	資料番号 an2k-005
テーマ	電子秤から RS-232C で送出されるデータを受信処理する		
使用機器	MPC-2100 (MPC-2000) 電子秤 : エーアンドディ EK-1200A		

■ 機器構成




■ 結線

MPC-2100



CNFG# 2 "2400b7pes1NONE"
INPUT# 2 EOL|10 TMOUT|2 RCV\$

MPC-2000 なら



CNFG# "2400b7pes1NONE"
INPUT# EOL|10 TMOUT|2 RCV\$

■ 概要

- 電子秤の RS-232C ポートから送出されるデータを受信・解析して FTMW の画面と MPC の LCD に表示します。

■ 電子秤から送出される文字列データ

例) WT,+000000.0 g<CR><LF>

- WT: ヘッダー文字。WT=安定時、US=不安定時、OL=オーバー
- +: 正負記号。負なら -。
- 000000.0 : データ。8 文字固定長、少数点は位置が変わったり、無かったりするかも?。
- g: 単位。
- 送出サイクル: 1 秒に 4 回弱垂れ流し。
- MPC 側からの制御は無し

■ このサンプルの壺

- INPUT#でタイムアウト付き LF 待ち受信。タイムアウトならリトライします。
- VAL () で受信文字列の中から重さの数値データを取り出します。
- SELSECT_CASE でヘッダーをチェック、想定外ならリトライします。

■ このサンプルの読み込みの基本形

```
CNFG# 2 "2400b7pes1NONE" /* 初期化
INPUT# 2 EOL|10 TMOUT|2 RCV$ /* LF まで受信 TMOUT 2 秒
W1=VAL (RCV$) /* RCV$ 中の最初の数値文字列
W2=VAL (0) /* " 次の "
PRINT W1 W2
5 6 /* 5.6g のとき
```

■ サンプルプログラム

```

CNFG# 2 "2400b7pes1NONE" /* 初期化
FORMAT
TOTAL_CNT=0 /* トータルカウント
RETRY_CNT=0 /* リトライカウント
DO
*RETRY
INPUT# 2 EOL|10 TMOUT|2 RCV$ /* LF まで受信 TMOUT 2 秒
IF rse_<>0 THEN /* TMOUT 処理。rse_は予約変数。
RETRY_CNT=RETRY_CNT+1
PRINT "tmout retry" RETRY_CNT rse_
GOTO *RETRY
END_IF
PR RCV$
ptr =RCV$ /* 文字列 RCV$のポインタ。ptr_は予約変数。
HEADER$=PTR$(2) /* RCV$の先頭から 2 文字を HEADER$へコピー
ptr =RCV$+14 /* ポインタを 14 文字進める
UNIT$=PTR$(1) /* ポインタ位置から 1 文字を UNIT$へコピー
SELECT_CASE HEADER$ /* ヘッダーをチェックする
CASE "WT" : RESULT$="○"
CASE "US" : RESULT$="△"
CASE "OL" : RESULT$="×"
CASE_ELSE /* 想定外
PRINT "invalid header"
GOTO *RETRY
END_SELECT
TOTAL_CNT=TOTAL_CNT+1
WEIGHT1$=STR$(VAL(RCV$)) /* 文字列 RCV$の中の最初の数値(この場合整数部)
SERCH RCV$ "." /* 秤の設定によっては小数部が無い場合がある(仮定)
IF ptr <>0 THEN /* 検索文字が有った場合 = 小数部が有った場合。
WEIGHT2$="."+STR$(VAL(0)) /* 文字列 RCV$の中の次の数値(この場合小数部)
ELSE /* ↑LCD 表示の都合で"."を","に替えてます
WEIGHT2$="" /* 小数部が無ければ空にする
END_IF
PRINT TOTAL_CNT RETRY_CNT RESULT$ WEIGHT1$ WEIGHT2$ UNIT$ /* FTMW 表示
BUF$=HEADER$+WEIGHT1$+WEIGHT2$+"G" /* LCD には小英字は表示できない
PR_LCD BUF$ /* MPC-2100 LCD 表示 例 "WT117,3G"
LOOP

```

実行結果 (FTMW 表示)

```

1 0 ○ 0 , 0 g
2 0 ○ 0 , 0 g
3 0 ○ 0 , 0 g
4 0 ○ 0 , 0 g
5 0 ○ 0 , 0 g
6 0 ○ 0 , 0 g
7 0 ○ 0 , 0 g
8 0 ○ 0 , 0 g
9 0 △ 51 , 3 g ←秤に物を乗せる
10 0 △ 111 , 9 g
11 0 △ 117 , 0 g
12 0 △ 117 , 2 g
13 0 △ 117 , 3 g
14 0 △ 117 , 3 g
15 0 △ 117 , 3 g
16 0 △ 117 , 3 g
17 0 ○ 117 , 3 g
18 0 ○ 117 , 3 g
19 0 ○ 117 , 3 g
20 0 ○ 117 , 3 g

```

以上